
TITLE: 
Reac05: propagation of steam explosion pressure waves in the secondary containment 
of a nuclear reactor. Similar to Stea05 but with reinforced concrete material model for 
the building. 
PROBLEM: 
This problem was suggested by ISMES (I) in the early-nineties. It represents the 
effects of a steam or gas explosion on the secondary containment building of a nuclear 
reactor. The explosion takes place in a liquid water pool located immediately below 
the primary containment (reactor vessel) and produces strong pressure waves, which 
propagate across the atmosphere in the reactor building, interacting with its walls and 
with some schematized internal structures. The explosive products are schematized by 
a high-pressure gas. 
 MESH: 
The model is 2D axisymmetric and uses 24 elements FL23 and 957 elements FL24 to 
represent the fluid, and 88 shell elements ED01 for the building walls and internals. 
Shell elements are made of 5 layers each, 3 of which representing the concrete and 2 
for the steel reinforcements. The calculation is ALE and starts with a hot and high-
pressure steam bubble in the lower part of the building. 
MATERIALS: 
The bottom of the building is assumed to be rigid, while the walls and internals, 
which are made of reinforced concrete material, are represented by the DPSF model, a 
Drucker-Prager constitutive law with softening and viscoplastic regularization, for the 
concrete, and by the VMSF (von Mises elasto-plastic material with softening) for the 
steel reinforcement. 
BOUNDARY CONDITIONS: 
The fluid-structure interaction along the deformable building walls is automaticcaly 
modelled by the FSA directive. 

LOADING: 
The system is initially at rest, but not in equilibrium. 
CALCULATION: 
The calculation is performed up to 250 ms. This time is sufficient to see propagation 
of the pressure waves to the top of the building and several reflections. 
RESULTS: 
Although no comparison with experiments can be done on this problem, the results 
appear physically correct. 
POST-TREATMENT 
Several animations of the computed results from this calculation are available on the 
EUROPLEXUS Consortium Web site. 
 
REFERENCES: 
This calculation is detailed in: 
1) F. Casadei: "Simulation of a Gas Explosion in a Reactor Containment by PLEXIS-
3C." Technical Note N. I.93.148, November 1993. 
2) J.J. Lopez Cela, F. Casadei and P. Pegon: "Fast Transient Analysis of Thin Shell 
Reinforced Concrete Structures with Drucker-Prager Model." SMiRT-14 Conference, 
Lyon (F), August 17-22, 1997. 
(both available on the EUROPLEXUS Consortium Web site). 
 
Numerical Solution 
 
REAC05 
 
The mesh generation file (K2000) is: 
 
*%siz 100 
* 
opti echo 1; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxpdroit.proc'; 
* 
opti titr 'REAC - 05'; 
opti dime 2 elem qua4; 
* 
p0=0 0; 
p1=0.00 6.32; 
p2=2.35 6.32; 
p3=2.35 14.42; 
p4=4.23 14.42;p4p=p4 plus p0;p4i=p4 plus p0; 
p5=4.80 14.42; 
p6=4.80 12.72; 
p7=14.76 12.72; 
p8=14.76 14.42; 
p9=20.35 14.42;p9i=p9 plus p0; 
p10=4.23 25.45;p10p=p10 plus p0;p10i=p10 plus p0; 
p11=20.35 25.45;p11p=p11 plus p0;p11i=p11 plus p0; 
p12=20.35 55.14;p12p=p12 plus p0; 
p13=0.00 67.02;p13p=p13 plus p0; 
p14=0.00 55.14; 
p15=0.00 25.45; 
p16=0.00 14.42; 
* 
tol=0.01; 
* 
c1=p1 d 4 p2; 
c2=p2 d 8 p3; 
c3=p3 d 4 p16; 
c4=p16 d 8 p1; 
boll=daller c1 c2 c3 c4 plan; 
* 
c1=p16 d 4 p3 d 3 p4; 
c2=p4 d 10 p10; 
c3=p10 d 7 p15; 
c4=p15 d 10 p16; 
flu1=daller c1 c2 c3 c4 plan; 
* 
c1=p15 d 7 p10 d 13 p11; 
c2=p11 d 25 p12; 
c3=p12 d 20 p14; 
c4=p14 d 25 p15; 
flu2=daller c1 c2 c3 c4 plan; 
* 
ella=20.35; 
ellb=11.87; 
rapp=ella / ellb; 
cir=cerc 30 (p14 plus (ellb 0)) p14 
            (p14 plus (0 ellb)); 
ell=cir affi rapp p14 p12;  
* 
p12d=ell poin proc p12; 
conf p12d p12; 
p13d=ell poin proc p13; 
conf p13d p13; 
* 
co=(p14 d 20 p12) et ell et (p13 d 10 p14); 
* 
* it is necessary to eliminate double points 
* in co so that it is seen as a closed contour 
* by operator surf 
* 
elim tol co; 
flu3=surf co plan; 
* 
str1=p4p d 10 p10p; 
str2=p10p d 13 p11p; 
str3=p9 d 10 p11p; 
str4=p11p d 25 p12p; 

str5=ell plus p0; 
grosso=(str5 elem 1) et (str5 elem 2) et 
       (str5 elem 3) et (str5 elem 4); 
sottile=diff str5 grosso; 
p12pd=str5 poin proc p12p; 
conf p12pd p12p; 
p13pd=str5 poin proc p13p; 
conf p13pd p13p; 
* 
flui=flu1 et flu2 et flu3; 
elim tol (boll et flui); 
stru=str1 et str2 et str3 et str4 et str5; 
elim tol stru; 
* 
c1=p4i d 1 p5 d 7 p8 d 5 p9i; 
c2=p9i d 10 p11i; 
c3=p11i d 13 p10i; 
c4=p10i d 10 p4i; 
flu4a=daller c1 c2 c3 c4 plan; 
c1=p5 d 2 p6; 
c2=p6 d 7 p7; 
c3=p7 d 2 p8; 
c4=p8 d 7 p5; 
flu4b=daller c1 c2 c3 c4 plan;  
flu4=flu4a et flu4b; 
elim tol flu4; 
* 
flui=flui et flu4; 
flut=flui elem 'TRI3'; 
fluq=flui elem 'QUA4'; 
mesh=boll et flut et fluq et stru; 
* 
simx=mesh poin droi p1 p13 tol; 
blocy=boll poin droi p1 p2 tol; 
blocy=blocy et (pxpdroit flu1 p3 p4 tol); 
blocy=blocy et (pxpdroit flu4 p4i p5 tol); 
blocy=blocy et (pxpdroit flu4 p6 p7 tol); 
blocy=blocy et (pxpdroit flu4 p8 p9i tol); 
blocy=blocy et p4p et p9; 
blocx=boll poin droi p2 p3 tol; 
blocx=blocx et (pxpdroit flu4 p5 p6 tol); 
blocx=blocx et (pxpdroit flu4 p7 p8 tol); 
blocx=blocx et p4p et p9; 
blocr=p4p et p9; 
* 
fsa=flu1 poin droi p4 p10 tol; 
fsa=fsa et (pxpdroit flu2 p10 p11 tol); 
fsa=fsa et (flu2 poin droi p11 p12 tol); 
fsa=fsa et (chan poi1 ell); 
fsa=fsa et (flu4 poin droi p4i p10i tol); 
fsa=fsa et (flu4 poin droi p10i p11i tol); 
fsa=fsa et (flu4 poin droi p11i p9i tol); 
* 
tn1=str1 poin proc ((p4p plus p10p) / 2); 
tn2=p10p; 
tn3=str2 poin proc ((p10p plus p11p) / 2); 
tn4=p11p; 
tn5=str3 poin proc ((p9 plus p11p) / 2); 
tn6=str4 poin proc ((p11p plus p12p) / 2); 
tn7=p12p; 
elmid=10.175 65.42; 
tn8=str5 poin proc elmid; 
tn9=p13p; 
* 
k=boll poin proc ((p1 plus p3) / 2); 
te1=boll elem 'CONTENENT' k; 
te1=te1 elem 1; 
k=flu1 poin proc ((p16 plus p10) / 2); 
te2=flu1 elem 'CONTENENT' k; 
te2=te2 elem 1; 
k=flu2 poin proc ((p15 plus p14) / 2); 

te3=flu2 elem 'CONTENENT' k; 
te3=te3 elem 1; 
k=flu2 poin proc ((p15 plus p12) / 2); 
te4=flu2 elem 'CONTENENT' k; 
te4=te4 elem 1; 
k=flu2 poin proc ((p11 plus p12) / 2); 
te5=flu2 elem 'CONTENENT' k; 
te5=te5 elem 1; 
k=p14; 
te6=flu2 elem 'CONTENENT' k; 
te6=te6 elem 1; 
k=flu2 poin proc ((p14 plus p12) / 2); 
te7=flu2 elem 'CONTENENT' k; 
te7=te7 elem 1; 
k=p12; 
te8=flu2 elem 'CONTENENT' k; 
te8=te8 elem 1; 
k=flu3 poin proc elmid; 
te9=flu3 elem 'CONTENENT' k; 
te9=te9 elem 1; 
k=p13; 
te10=flu3 elem 'CONTENENT' k; 
te10=te10 elem 1; 
k=str1 poin proc ((p4p plus p10p) / 2); 
te11=str1 elem 'CONTENENT' k; 
te11=te11 elem 1; 
k=str2 poin proc ((p10p plus p11p) / 2); 
te12=str2 elem 'CONTENENT' k; 
te12=te12 elem 1; 
k=str3 poin proc ((p9 plus p11p) / 2); 
te13=str3 elem 'CONTENENT' k; 
te13=te13 elem 1; 
k=p11p; 
te14=str3 elem 'CONTENENT' k; 
te14=te14 elem 1; 
te15=str2 elem 'CONTENENT' k; 
te15=te15 elem 1; 
te16=str4 elem 'CONTENENT' k; 
te16=te16 elem 1; 
k=str4 poin proc ((p11p plus p12p) / 2); 
te17=str4 elem 'CONTENENT' k; 
te17=te17 elem 1; 
k=p12p; 

te18=str4 elem 'CONTENENT' k; 
te18=te18 elem 1; 
k=str5 poin proc elmid; 
te19=str5 elem 'CONTENENT' k; 
te19=te19 elem 1; 
k=p13p; 
te20=str5 elem 'CONTENENT' k; 
te20=te20 elem 1; 
tpln=tn1 et tn2 et tn3 et tn4 et tn5 et tn6 
         et tn7 et tn8 et tn9; 
tple=te1 et te2 et te3 et te4 et te5 et te6 
         et te7 et te8 et te9 et te10 et te11 
         et te12 et te13 et te14 et te15 
         et te16 et te17 et te18 et te19 
         et te20; 
tpln1=tn3 et tn6; 
tple1=te20 et te17 et te12; 
xpln=boll poin droi p1 p2 tol; 
xpln=xpln et (boll poin droi p2 p3 tol); 
xpln=xpln et (pxpdroit flu1 p3 p4 tol); 
xpln=xpln et (pxpdroit flu1 p4 p10 tol); 
xpln=xpln et (pxpdroit flu2 p10 p11 tol); 
xpln=xpln et (pxpdroit flu2 p11 p12 tol); 
xpln=xpln et ell; 
xpln=chan poi1 xpln; 
* 
ss=chan poi1 stru; 
lag=ss et p1 et p2 et p3 et p5 
       et p6 et p7 et p8; 
* 
mesh=boll et flut et fluq et stru 
     et simx et blocx et blocy et blocr 
     et lag et fsa et tpln et tple 
     et xpln et tpln1 et tple1; 
tass mesh; 
 
opti sauv form 'vl_jrc_reac05.msh'; 
sauv form mesh; 
* 
opti trac psc ftra 'vl_jrc_reac05_mesh.ps'; 
trac mesh; 
fin; 

The EUROPLEXUS input file reads: 
 
REAC - 05  
$ dpsf material, ed01 with layers 
 ECHO 
*CONV win 
CAST MESH 
AXIS NONL ALE 
$ 
DIME 
 PT3L 89 PT2L 1088 
 FL23 24 FL24 957 ED01 88 ZONE 3 
 NALE 43 NBLE 1088 
 PRES 65 MTEL 88  
 ELVC 186 MTPO 94  
 ECRO 28821 
TERM 
$ 
GEOM 
  FL23 FLUT FL24 FLUQ BOLL 
  ED01 STRU 
TERM 
$ 
COMP 
  EPAI 1.15 LECT STR1 TERM 
       0.61 LECT STR2 TERM 
       0.90 LECT STR3 STR4 SOTTILE TERM 
       1.75 LECT GROSSO TERM 
  LAYE 5 FRAC 0.1 0.01 0.78 0.01 0.1 
         NGPZ 1 1 2 1 1 LECT STRU TERM 
$ 
GRIL LAGR LECT LAG TERM 
     EULE LECT FSA TERM 
     ALE  LECT FLUI TERM 
     AUTO AUTR 
$ 
$ concrete (softening) 
MATE DPSF RO 2.4E3 YOUN 2.E10 NU 0.2 ALF1 1.299 
     C1 5.7735E6 BETA 0.20 ETA 7.E-5 
     TRAA 3 1.299 0.0 1.299 5.E-4 1.299 5.E+2 
     TRAC 3 5.7735e6 0.0 1.44338E6 5.15E-3 1.44338E6 5.E+2 
     LECT STRU TERM LAYE LECT 1 3 5 TERM  
$ 
$ steel (elastoplastic) 
     VMSF RO 7.8E3 YOUN 2.1E11 NU 0.3 ELAS 6.8E8 
     ETA 0. TRAC 2 6.8E8 3.23810E-03 6.8E8 5.E+2  
     LECT STRU TERM LAYE LECT 2 4 TERM 
$ high-pressure perfect gas 
     FLUT RO 111.50 EINT 12.764E5 GAMM 1.4 PB 0 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 LECT BOLL TERM 
$ same gas at a lower pressure 

     FLUT RO 1.2 EINT 2.0833E5 GAMM 1.4 PB 0 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 LECT FLUI TERM 
$ 
CHAR 1 FACT 2 PRES COQU 1.E5 
              LECT STR3 STR4 STR5 TERM 
              TABL 2 0. 1. 100. 1. 
$ 
link coup 
     BLOQ 1 LECT BLOCX TERM 
          2 LECT BLOCY TERM 
          3 LECT BLOCR TERM 
     CONT SPLA NX 1 NY 0 LECT SIMX TERM 
     FSA  LECT FSA TERM 
$ 
ECRI DEPL VITE ECRO TFRE 25.E-3 
          POIN LECT TPLN TERM 
          ELEM LECT TPLE TERM 
     FICH ALIC TFRE 2.5E-3 
     FICH ALIC TEMP TFRE 2.5E-4 
               POIN LECT TPLN TERM 
               ELEM LECT TPLE TERM 
     FICH K200 TFRE 10.E-3 POIN TOUS 
               VARI DEPL VITE ECRO ECRC LECT 1 2 4 8 TERM 
$ 
OPTI NOTE 
     log 1 
     REZO GAM0 0.5 
CALCUL TINI 0 TEND 250.E-3 
*======================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1000.0 'Time [ms]'  
* 
COUR  1 'dy_n3' DEPL COMP 2 NOEU LECT tn3 TERM 
COUR  2 'dy_n9' DEPL COMP 2 NOEU LECT tn9 TERM 
* 
trac 1 2 axes 1.0 'DISPL. [M]' 
* 
QUAL DEPL COMP 2 LECT tn3 TERM REFE -9.17858E-1 TOLE 5.E-2 
     DEPL COMP 2 LECT tn9 TERM REFE  2.25577E-1 TOLE 5.E-2 
*======================================================================= 
FIN 

 
Some results: final pressure distribution and final structure deformation: 

Intermediate structure velocities and final fluid velocities: 
 

 
 
 
Comparison of elastic and concrete (elastoplastic) material solutions: 
 

 


